Zernike Polynomials

1 Introduction

Often, to aid in the interpretation of optical test results it is convenient to express wavefront data in polynomial form. Zernike polynomials are often used for this purpose since they are made up of terms that are of the same form as the types of aberrations often observed in optical tests (Zernike, 1934). This is not to say that Zernike polynomials are the best polynomials for fitting test data. Sometimes Zernike polynomials give a poor representation of the wavefront data. For example, Zernikes have little value when air turbulence is present. Likewise, fabrication errors in the single point diamond turning process cannot be represented using a reasonable number of terms in the Zernike polynomial. In the testing of conical optical elements, additional terms must be added to Zernike polynomials to accurately represent alignment errors. The blind use of Zernike polynomials to represent test results can lead to disastrous results.

Zernike polynomials are one of an infinite number of complete sets of polynomials in two variables, ρ and θ, that are orthogonal in a continuous fashion over the interior of a unit circle. It is important to note that the Zernikes are orthogonal only in a continuous fashion over the interior of a unit circle, and in general they will not be orthogonal over a discrete set of data points within a unit circle.

Zernike polynomials have three properties that distinguish them from other sets of orthogonal polynomials. First, they have simple rotational symmetry properties that lead to a polynomial product of the form

$$r[\rho]\, g[\theta],$$

where g[θ] is a continuous function that repeats self every 2π radians and satisfies the requirement that rotating the coordinate system by an angle α does not change the form of the polynomial. That is

$$g[\theta + \alpha] = g[\theta]\, g[\alpha].$$

The set of trigonometric functions

$$g[\theta] = e^{\pm i m \theta},$$

where m is any positive integer or zero, meets these requirements.

The second property of Zernike polynomials is that the radial function must be a polynomial in ρ of degree 2n and contain no power of ρ less than m. The third property is that r[ρ] must be even if m is even, and odd if m is odd.

The radial polynomials can be derived as a special case of Jacobi polynomials, and tabulated as $r[n, m, \rho]$. Their orthogonality and normalization properties are given by

$$\int_0^1 r[n, m, \rho]\, r[n', m, \rho]\, \rho\, d\rho = \frac{1}{2(n+1)} \text{KroneckerDelta}[n - n']$$

and

$r[n, m, 1] = 1.$

As stated above, r[n, m, ρ] is a polynomial of order 2n and it can be written as

$$\texttt{r[n\_, m\_, }\rho\texttt{\_] :=} \sum_{s=0}^{n-m} (-1)^s \frac{(2n-m-s)!}{s!\,(n-s)!\,(n-m-s)!} \rho^{2(n-s)-m}$$

In practice, the radial polynomials are combined with sines and cosines rather than with a complex exponential. It is convenient to write

```
rcos[n_, m_, ρ_] := r[n, m, ρ] Cos[m θ]
```

and

```
rsin[n_, m_, ρ_] := r[n, m, ρ] Sin[m θ]
```

The final Zernike polynomial series for the wavefront opd Δw can be written as

$$\Delta w[\rho\_,\ \theta\_] := \overline{\Delta w} + \sum_{n=1}^{nmax} \left(a[n]\ r[n, 0, \rho] + \sum_{m=1}^{n} (b[n, m]\ rcos[n, m, \rho] + c[n, m]\ rsin[n, m, \rho]) \right)$$

where Δw[ρ, θ] is the mean wavefront opd, and a[n], b[n,m], and c[n,m] are individual polynomial coefficients. For a symmetrical optical system, the wave aberrations are symmetrical about the tangential plane and only even functions of θ are allowed. In general, however, the wavefront is not symmetric, and both sets of trigonometric terms are included.

2 Calculating Zernikes

For the example below the degree of the Zernike polynomials is selected to be 6. The value of nDegree can be changed if a different degree is desired.

The array zernikePolar contains Zernike polynomials in polar coordinates (ρ, θ), while the array zernikeXy contains the Zernike polynomials in Cartesian, (x, y), coordinates. zernikePolarList and zernikeXyList contains the Zernike number in column 1, the n and m values in columns 2 and 3, and the Zernike polynomial in column 4.

```
nDegree = 6;

i = 0;
Do[If[m == 0, {i = i + 1, temp[i] = {i - 1, n, m, r[n, m, ρ]}},
            {i = i + 1, temp[i] = {i - 1, n, m, Factor[rcos[n, m, ρ]]},
    i = i + 1, temp[i] = {i - 1, n, m, Factor[rsin[n, m, ρ]]}}], {n, 0, nDegree}, {m, n, 0, -1}];
```

```
zernikePolarList = Array[temp, i];
Clear[temp];
Do[zernikePolar[i - 1] = zernikePolarList[[i, 4]], {i, 1, Length[zernikePolarList]}];
```

$$\texttt{zernikeXyList = Map[TrigExpand, zernikePolarList] /. } \left\{\rho \to \sqrt{x^2+y^2},\ \texttt{Cos}[\theta] \to \frac{x}{\sqrt{x^2+y^2}},\ \texttt{Sin}[\theta] \to \frac{y}{\sqrt{x^2+y^2}}\right\};$$

```
Do[zernikeXy[i - 1] = zernikeXyList[[i, 4]], {i, 1, Length[zernikeXyList]}]
```

2.1 Tables of Zernikes

In the tables term # 1 is a constant or piston term, while terms # 2 and # 3 are tilt terms. Term # 4 represents focus. Thus, terms # 2 through # 4 represent the Gaussian or paraxial properties of the wavefront. Terms # 5 and # 6 are astigmatism plus defocus. Terms # 7 and # 8 represent coma and tilt, while term # 9 represents third-order spherical and focus. Likewise terms # 10 through # 16 represent fifth-order aberration, terms # 17 through # 25 represent seventh-order aberrations, terms # 26 through # 36 represent ninth-order aberrations, and terms # 37 through # 49 represent eleventh-order aberrations.

Each term contains the appropriate amount of each lower order term to make it orthogonal to each lower order term. Also, each term of the Zernikes minimizes the rms wavefront error to the order of that term. Adding other aberrations of lower order can only increase the rms error. Furthermore, the average value of each term over the unit circle is zero.

2.1.1 Zernikes in polar coordinates

```
TableForm[zernikePolarList, TableHeadings -> {{}, {"#", "n", "m", "Polynomial"}}]
```

#	n	m	Polynomial
0	0	0	1
1	1	1	$\rho\ \mathrm{Cos}[\theta]$
2	1	1	$\rho\ \mathrm{Sin}[\theta]$
3	1	0	$-1 + 2\rho^2$
4	2	2	$\rho^2\ \mathrm{Cos}[2\theta]$
5	2	2	$\rho^2\ \mathrm{Sin}[2\theta]$
6	2	1	$\rho\ (-2 + 3\rho^2)\ \mathrm{Cos}[\theta]$
7	2	1	$\rho\ (-2 + 3\rho^2)\ \mathrm{Sin}[\theta]$
8	2	0	$1 - 6\rho^2 + 6\rho^4$
9	3	3	$\rho^3\ \mathrm{Cos}[3\theta]$
10	3	3	$\rho^3\ \mathrm{Sin}[3\theta]$

11	3	2	$\rho^2 (-3 + 4 \rho^2) \text{Cos}[2 \theta]$
12	3	2	$\rho^2 (-3 + 4 \rho^2) \text{Sin}[2 \theta]$
13	3	1	$\rho (3 - 12 \rho^2 + 10 \rho^4) \text{Cos}[\theta]$
14	3	1	$\rho (3 - 12 \rho^2 + 10 \rho^4) \text{Sin}[\theta]$
15	3	0	$-1 + 12 \rho^2 - 30 \rho^4 + 20 \rho^6$
16	4	4	$\rho^4 \text{Cos}[4 \theta]$
17	4	4	$\rho^4 \text{Sin}[4 \theta]$
18	4	3	$\rho^3 (-4 + 5 \rho^2) \text{Cos}[3 \theta]$
19	4	3	$\rho^3 (-4 + 5 \rho^2) \text{Sin}[3 \theta]$
20	4	2	$\rho^2 (6 - 20 \rho^2 + 15 \rho^4) \text{Cos}[2 \theta]$
21	4	2	$\rho^2 (6 - 20 \rho^2 + 15 \rho^4) \text{Sin}[2 \theta]$
22	4	1	$\rho (-4 + 30 \rho^2 - 60 \rho^4 + 35 \rho^6) \text{Cos}[\theta]$
23	4	1	$\rho (-4 + 30 \rho^2 - 60 \rho^4 + 35 \rho^6) \text{Sin}[\theta]$
24	4	0	$1 - 20 \rho^2 + 90 \rho^4 - 140 \rho^6 + 70 \rho^8$
25	5	5	$\rho^5 \text{Cos}[5 \theta]$
26	5	5	$\rho^5 \text{Sin}[5 \theta]$
27	5	4	$\rho^4 (-5 + 6 \rho^2) \text{Cos}[4 \theta]$
28	5	4	$\rho^4 (-5 + 6 \rho^2) \text{Sin}[4 \theta]$
29	5	3	$\rho^3 (10 - 30 \rho^2 + 21 \rho^4) \text{Cos}[3 \theta]$
30	5	3	$\rho^3 (10 - 30 \rho^2 + 21 \rho^4) \text{Sin}[3 \theta]$
31	5	2	$\rho^2 (-10 + 60 \rho^2 - 105 \rho^4 + 56 \rho^6) \text{Cos}[2 \theta]$
32	5	2	$\rho^2 (-10 + 60 \rho^2 - 105 \rho^4 + 56 \rho^6) \text{Sin}[2 \theta]$
33	5	1	$\rho (5 - 60 \rho^2 + 210 \rho^4 - 280 \rho^6 + 126 \rho^8) \text{Cos}[\theta]$
34	5	1	$\rho (5 - 60 \rho^2 + 210 \rho^4 - 280 \rho^6 + 126 \rho^8) \text{Sin}[\theta]$
35	5	0	$-1 + 30 \rho^2 - 210 \rho^4 + 560 \rho^6 - 630 \rho^8 + 252 \rho^{10}$
36	6	6	$\rho^6 \text{Cos}[6 \theta]$
37	6	6	$\rho^6 \text{Sin}[6 \theta]$
38	6	5	$\rho^5 (-6 + 7 \rho^2) \text{Cos}[5 \theta]$
39	6	5	$\rho^5 (-6 + 7 \rho^2) \text{Sin}[5 \theta]$
40	6	4	$\rho^4 (15 - 42 \rho^2 + 28 \rho^4) \text{Cos}[4 \theta]$
41	6	4	$\rho^4 (15 - 42 \rho^2 + 28 \rho^4) \text{Sin}[4 \theta]$
42	6	3	$\rho^3 (-20 + 105 \rho^2 - 168 \rho^4 + 84 \rho^6) \text{Cos}[3 \theta]$
43	6	3	$\rho^3 (-20 + 105 \rho^2 - 168 \rho^4 + 84 \rho^6) \text{Sin}[3 \theta]$
44	6	2	$\rho^2 (15 - 140 \rho^2 + 420 \rho^4 - 504 \rho^6 + 210 \rho^8) \text{Cos}[2 \theta]$
45	6	2	$\rho^2 (15 - 140 \rho^2 + 420 \rho^4 - 504 \rho^6 + 210 \rho^8) \text{Sin}[2 \theta]$

46	6	1	$\rho\,(-6+105\,\rho^2-560\,\rho^4+1260\,\rho^6-1260\,\rho^8+462\,\rho^{10})\,\text{Cos}[\theta]$
47	6	1	$\rho\,(-6+105\,\rho^2-560\,\rho^4+1260\,\rho^6-1260\,\rho^8+462\,\rho^{10})\,\text{Sin}[\theta]$
48	6	0	$1-42\,\rho^2+420\,\rho^4-1680\,\rho^6+3150\,\rho^8-2772\,\rho^{10}+924\,\rho^{12}$

2.1.2 Zernikes in Cartesian coordinates

```
TableForm[zernikeXyList, TableHeadings -> {{}, {"#", "n", "m", "Polynomial"}}]
```

#	n	m	Polynomial
0	0	0	1
1	1	1	x
2	1	1	y
3	1	0	$-1+2\,(x^2+y^2)$
4	2	2	x^2-y^2
5	2	2	$2\,x\,y$
6	2	1	$-2\,x+3\,x\,(x^2+y^2)$
7	2	1	$-2\,y+3\,y\,(x^2+y^2)$
8	2	0	$1-6\,(x^2+y^2)+6\,(x^2+y^2)^2$
9	3	3	$x^3-3\,x\,y^2$
10	3	3	$3\,x^2\,y-y^3$
11	3	2	$-3\,x^2+3\,y^2+4\,x^2\,(x^2+y^2)-4\,y^2\,(x^2+y^2)$
12	3	2	$-6\,x\,y+8\,x\,y\,(x^2+y^2)$
13	3	1	$3\,x-12\,x\,(x^2+y^2)+10\,x\,(x^2+y^2)^2$
14	3	1	$3\,y-12\,y\,(x^2+y^2)+10\,y\,(x^2+y^2)^2$
15	3	0	$-1+12\,(x^2+y^2)-30\,(x^2+y^2)^2+20\,(x^2+y^2)^3$
16	4	4	$x^4-6\,x^2\,y^2+y^4$
17	4	4	$4\,x^3\,y-4\,x\,y^3$
18	4	3	$-4\,x^3+12\,x\,y^2+5\,x^3\,(x^2+y^2)-15\,x\,y^2\,(x^2+y^2)$
19	4	3	$-12\,x^2\,y+4\,y^3+15\,x^2\,y\,(x^2+y^2)-5\,y^3\,(x^2+y^2)$
20	4	2	$6\,x^2-6\,y^2-20\,x^2\,(x^2+y^2)+20\,y^2\,(x^2+y^2)+15\,x^2\,(x^2+y^2)^2-15\,y^2\,(x^2+y^2)^2$
21	4	2	$12\,x\,y-40\,x\,y\,(x^2+y^2)+30\,x\,y\,(x^2+y^2)^2$
22	4	1	$-4\,x+30\,x\,(x^2+y^2)-60\,x\,(x^2+y^2)^2+35\,x\,(x^2+y^2)^3$
23	4	1	$-4\,y+30\,y\,(x^2+y^2)-60\,y\,(x^2+y^2)^2+35\,y\,(x^2+y^2)^3$
24	4	0	$1-20\,(x^2+y^2)+90\,(x^2+y^2)^2-140\,(x^2+y^2)^3+70\,(x^2+y^2)^4$
25	5	5	$x^5-10\,x^3\,y^2+5\,x\,y^4$
26	5	5	$5\,x^4\,y-10\,x^2\,y^3+y^5$
27	5	4	$-5\,x^4+30\,x^2\,y^2-5\,y^4+6\,x^4\,(x^2+y^2)-36\,x^2\,y^2\,(x^2+y^2)+6\,y^4\,(x^2+y^2)$
28	5	4	$-20\,x^3\,y+20\,x\,y^3+24\,x^3\,y\,(x^2+y^2)-24\,x\,y^3\,(x^2+y^2)$
29	5	3	$10\,x^3-30\,x\,y^2-30\,x^3\,(x^2+y^2)+90\,x\,y^2\,(x^2+y^2)+21\,x^3\,(x^2+y^2)^2-63\,x\,y^2\,(x^2+y^2)^2$
30	5	3	$30\,x^2\,y-10\,y^3-90\,x^2\,y\,(x^2+y^2)+30\,y^3\,(x^2+y^2)+63\,x^2\,y\,(x^2+y^2)^2-21\,y^3\,(x^2+y^2)^2$
31	5	2	$-10\,x^2+10\,y^2+60\,x^2\,(x^2+y^2)-60\,y^2\,(x^2+y^2)-105\,x^2\,(x^2+y^2)^2+105\,y^2\,(x^2+y^2)^2+56\,x^2\,(x^2+y^2)^3-56\,y^2\,(x^2+y^2)^3$
32	5	2	$-20\,x\,y+120\,x\,y\,(x^2+y^2)-210\,x\,y\,(x^2+y^2)^2+112\,x\,y\,(x^2+y^2)^3$
33	5	1	$5\,x-60\,x\,(x^2+y^2)+210\,x\,(x^2+y^2)^2-280\,x\,(x^2+y^2)^3+126\,x\,(x^2+y^2)^4$
34	5	1	$5\,y-60\,y\,(x^2+y^2)+210\,y\,(x^2+y^2)^2-280\,y\,(x^2+y^2)^3+126\,y\,(x^2+y^2)^4$
35	5	0	$-1+30\,(x^2+y^2)-210\,(x^2+y^2)^2+560\,(x^2+y^2)^3-630\,(x^2+y^2)^4+252\,(x^2+y^2)^5$
36	6	6	$x^6-15\,x^4\,y^2+15\,x^2\,y^4-y^6$
37	6	6	$6\,x^5\,y-20\,x^3\,y^3+6\,x\,y^5$
38	6	5	$-6\,x^5+60\,x^3\,y^2-30\,x\,y^4+7\,x^5\,(x^2+y^2)-70\,x^3\,y^2\,(x^2+y^2)+35\,x\,y^4\,(x^2+y^2)$
39	6	5	$-30\,x^4\,y+60\,x^2\,y^3-6\,y^5+35\,x^4\,y\,(x^2+y^2)-70\,x^2\,y^3\,(x^2+y^2)+7\,y^5\,(x^2+y^2)$
40	6	4	$15\,x^4-90\,x^2\,y^2+15\,y^4-42\,x^4\,(x^2+y^2)+252\,x^2\,y^2\,(x^2+y^2)-42\,y^4\,(x^2+y^2)+28\,x^4\,(x^2+y^2)^2-168\,x^2\,y^2\,(x^2+y^2)^2+28\,y^4\,(x^2+y^2)^2$

41	6	4	$60x^3y - 60xy^3 - 168x^3y(x^2+y^2) + 168xy^3(x^2+y^2) + 112x^3y(x^2+y^2)^2 - 112xy^3(x^2+y^2)^2$
42	6	3	$-20x^3 + 60xy^2 + 105x^3(x^2+y^2) - 315xy^2(x^2+y^2) - 168x^3(x^2+y^2)^2 + 504xy^2(x^2+y^2)^2 + 84x^3(x^2+y^2)^3 - 252xy^2(x^2+y^2)^3$
43	6	3	$-60x^2y + 20y^3 + 315x^2y(x^2+y^2) - 105y^3(x^2+y^2) - 504x^2y(x^2+y^2)^2 + 168y^3(x^2+y^2)^2 + 252x^2y(x^2+y^2)^3 - 84y^3(x^2+y^2)^3$
44	6	2	$15x^2 - 15y^2 - 140x^2(x^2+y^2) + 140y^2(x^2+y^2) + 420x^2(x^2+y^2)^2 - 420y^2(x^2+y^2)^2 - 504x^2(x^2+y^2)^3 + 504y^2(x^2+y^2)^3 + 210x^2(x^2+y^2)^4 - 210y^2(x^2+y^2)^4$
45	6	2	$30xy - 280xy(x^2+y^2) + 840xy(x^2+y^2)^2 - 1008xy(x^2+y^2)^3 + 420xy(x^2+y^2)^4$
46	6	1	$-6x + 105x(x^2+y^2) - 560x(x^2+y^2)^2 + 1260x(x^2+y^2)^3 - 1260x(x^2+y^2)^4 + 462x(x^2+y^2)^5$
47	6	1	$-6y + 105y(x^2+y^2) - 560y(x^2+y^2)^2 + 1260y(x^2+y^2)^3 - 1260y(x^2+y^2)^4 + 462y(x^2+y^2)^5$
48	6	0	$1 - 42(x^2+y^2) + 420(x^2+y^2)^2 - 1680(x^2+y^2)^3 + 3150(x^2+y^2)^4 - 2772(x^2+y^2)^5 + 924(x^2+y^2)^6$

2.2 OSC Zernikes

Much of the early work using Zernike polynomials in the computer analysis of interferograms was performed by John Loomis at the Optical Sciences Center, University of Arizona in the 1970s. In the OSC work Zernikes for n= 1 through 5 and the n=6, m=0 term were used. The n=m=0 term (piston term) was used in interferogram analysis, but it was not included in the numbering of the Zernikes. Thus, there were 36 Zernike terms, plus the piston term used.

3 Zernike Plots

A few sample plots are given in this section. More plots can be found at http://www.optics.arizona.edu/jcwyant/Zernikes/ZernikePolynomials.htm.

3.1 Density Plots

```
zernikeNumber = 8;

temp = zernikeXy[zernikeNumber];
DensityPlot[If[x^2 + y^2 ≤ 1, (Cos[2 π temp])^2, 1], {x, -1, 1}, {y, -1, 1},
  PlotLabel → "Zernike #" <> ToString[zernikeNumber], ColorFunction → GrayLevel, PlotPoints -> 150, Mesh -> False];
```



3.2 3D Plots

```
zernikeNumber = 8;
```

```
temp = zernikeXy[zernikeNumber]; Graphics3D[Plot3D[
  {If[x^2 + y^2 ≤ 1, temp, 1], If[x^2 + y^2 ≤ 1, Hue[temp, 1, 1], Hue[1, 1, 1]]}, {x, -1, 1}, {y, -1, 1}, PlotPoints → 40]];
```



```
zernikeNumber = 8;

temp = zernikeXy[zernikeNumber];
Graphics3D[Plot3D[If[x^2 + y^2 ≤ 1, temp, 1], {x, -1, 1}, {y, -1, 1}, PlotPoints → 40,
    LightSources -> {{{1., 0., 1.}, RGBColor[1, 0, 0]},
     {{1., 1., 1.}, RGBColor[0, 1, 0]}, {{0., 1., 1.}, RGBColor[0, 0, 1]},
           {{-1., 0., -1.}, RGBColor[1, 0, 0]}, {{-1., -1., -1.}, RGBColor[0, 1, 0]},
     {{0., -1., -1.}, RGBColor[0, 0, 1]}}]];
```



3.3 Cylindrical Plot 3D

```
zernikeNumber = 8;

temp = zernikePolar[zernikeNumber];
gr = CylindricalPlot3D[temp, {ρ, 0, 1}, {θ, 0, 2 π}, BoxRatios → {1, 1, 0.5}, Boxed → False, Axes → False];
```



```
zernikeNumber = 5;

temp = zernikePolar[zernikeNumber];
gr = CylindricalPlot3D[{temp, Hue[temp]}, {ρ, 0, 1},
   {θ, 0, 2 π}, BoxRatios → {1, 1, 0.5}, Boxed → False, Axes → False, Lighting → False];
```



Can rotate without getting dark side

```
zernikeNumber = 5;

temp = zernikePolar[zernikeNumber];
gr = CylindricalPlot3D[temp, {ρ, 0, 1}, {θ, 0, 2 π}, BoxRatios → {1, 1, 0.5},
   Boxed → False, Axes → False, LightSources -> {{{1., 0., 1.}, RGBColor[1, 0, 0]},
     {{1., 1., 1.}, RGBColor[0, 1, 0]}, {{0., 1., 1.}, RGBColor[0, 0, 1]},
            {{-1., 0., -1.}, RGBColor[1, 0, 0]}, {{-1., -1., -1.}, RGBColor[0, 1, 0]},
     {{0., -1., -1.}, RGBColor[0, 0, 1]}}];
```



```
zernikeNumber = 16;

temp = zernikePolar[zernikeNumber];
gr = CylindricalPlot3D[temp, {ρ, 0, 1}, {θ, 0, 2 π}, BoxRatios → {1, 1, 0.5},
   Boxed → False, Axes → False, LightSources -> {{{1., 0., 1.}, RGBColor[1, 0, 0]},
     {{1., 1., 1.}, RGBColor[.5, 1, 0]}, {{0., 1., 1.}, RGBColor[1, 0, 0]},
            {{-1., 0., -1.}, RGBColor[1, 0, 0]}, {{-1., -1., -1.}, RGBColor[.5, 1, 0]},
     {{0., -1., -1.}, RGBColor[1, 0, 0]}}];
```



3.4 Surfaces of Revolution

```
zernikeNumber = 8;

temp = zernikePolar[zernikeNumber];
SurfaceOfRevolution[temp, {ρ, 0, 1}, PlotPoints → 40, BoxRatios → {1, 1, 0.5},
    LightSources -> {{{1., 0., 1.}, RGBColor[1, 0, 0]},
    {{1., 1., 1.}, RGBColor[.5, 1, 0]}, {{0., 1., 1.}, RGBColor[1, 0, 0]},
            {{-1., 0., -1.}, RGBColor[1, 0, 0]}, {{-1., -1., -1.}, RGBColor[.5, 1, 0]},
    {{0., -1., -1.}, RGBColor[1, 0, 0]}}];
```



3.5 3D Shadow Plots

```
zernikeNumber = 5;
```

```
temp = zernikeXy[zernikeNumber];
ShadowPlot3D[temp, {x, -√(1 - y²), √(1 - y²)}, {y, -1, 1}, PlotPoints → 40];
```



3.6 Animated Plots

3.6.1 Animated Density Plots

```
zernikeNumber = 3;

temp = zernikeXy[zernikeNumber];
MovieDensityPlot[If[x^2 + y^2 < 1, Sin[(temp + t y^2) π]^2, 1], {x, -1, 1}, {y, -1, 1},
  {t, -7, 4, 1}, PlotPoints → 100, Mesh -> False, FrameTicks -> None, Frame → False];
```



3.6.2 Animated 3D Shadow Plots

```
zernikeNumber = 5;
```

```
temp = zernikeXy[zernikeNumber];
g = ShadowPlot3D[temp, {x, -√(1 - y²), √(1 - y²)}, {y, -1, 1}, PlotPoints → 40, DisplayFunction → Identity];
SpinShow[ g, Frames -> 6,
  SpinRange -> {0 Degree, 360 Degree} ]
```



3.6.3 Animated Cylindrical Plot 3D

```
zernikeNumber = 5;

temp = zernikePolar[zernikeNumber];
gr = CylindricalPlot3D[{temp, Hue[Abs[temp + .4]]}, {ρ, 0, 1}, {θ, 0, 2 π},
   BoxRatios → {1, 1, 0.5}, Boxed → False, Axes → False, Lighting → False, DisplayFunction → Identity];
SpinShow[ gr, Frames -> 6,
  SpinRange -> {0 Degree, 360 Degree} ]
```



3.7 Two pictures stereograms

```
zernikeNumber = 8;

Print["Zernike #" <> ToString[zernikeNumber]];
ed = 0.6;
temp = zernikeXy[zernikeNumber];
f[x_, y_] := temp /; (x^2 + y^2) < 1
f[x_, y_] := 1 /; (x^2 + y^2) >= 1
plottemp = Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, Boxed → False,
   Axes → False, DisplayFunction → Identity, PlotPoints → 50, Mesh -> False];
Show[GraphicsArray[{Show[plottemp, ViewPoint -> {-ed / 2, -2.4, 2.}, ViewCenter → 0.5 + {-ed / 2, 0, 0}],
    Show[plottemp, ViewPoint -> {ed / 2, -2.4, 2.}, ViewCenter → 0.5 + {ed / 2, 0, 0}]},
   GraphicsSpacing → 0], PlotLabel → zernikeXy[zernikeNumber]];
```

```
Zernike #8
```

$$1 - 6 (x^2 + y^2) + 6 (x^2 + y^2)^2$$



3.8 Single picture stereograms

```
zernikeNumber = 8;

temp = zernikeXy[zernikeNumber];
tempPlot = Plot3D[If[x^2 + y^2 < 1, temp, 1], {x, -1, 1}, {y, -1, 1}, PlotPoints → 400, DisplayFunction → Identity];
SIRDS[tempPlot];
Clear[temp, tempPlot];
```



4 Relationship between Zernike polynomials and third-order aberrations

4.1 Wavefront aberrations

The third-order wavefront aberrations can be written as shown in the table below. Because there is no field dependence in these terms they are not true Seidel aberrations. Wavefront measurement using an interferometer only provides data at a single field point. This causes field curvature to look like focus and distortion to look like tilt. Therefore, a number of field points must be measured to determine the Seidel aberrations.

```
thirdOrderAberration = {{"piston", w00}, {"tilt", w11 ρ Cos[θ - αTilt]}, {"focus", w20 ρ^2},
   {"astigmatism", w22 ρ^2 Cos[θ - αAst]^2}, {"coma", w31 ρ^3 Cos[θ - αComa]}, {"spherical", w40 ρ^4}};
```

```
TableForm[thirdOrderAberration]
```

piston	w_0
tilt	$\rho \text{Cos}[\theta - \alpha_{Tilt}]\, w_{11}$
focus	$\rho^2 w_{20}$
astigmatism	$\rho^2 \text{Cos}[\theta - \alpha_{Ast}]^2 w_{22}$
coma	$\rho^3 \text{Cos}[\theta - \alpha_{Coma}]\, w_{31}$
spherical	$\rho^4 w_{40}$

4.2 Zernike terms

First-order wavefront properties and third-order wavefront aberration coefficients can be obtained from the Zernike polynomials. Let the coefficients of the first nine zernikes be given by

```
zernikeCoefficient = {z0, z1, z2, z3, z4, z5, z6, z7, z8};
```

The coefficients can be multiplied times the Zernike polynomials to give the wavefront aberration.

```
wavefrontAberrationList = Table[zernikeCoefficient zernikePolarList[[Range[1, 9], 4]]];
```

We will now express the wavefront aberrations and the corresponding Zernike terms in a table

4.3 Table of Zernikes and aberrations

```
wavefrontAberrationLabels = {"piston", "x-tilt", "y-tilt", "focus", "astigmatism at 0 degrees & focus",
   "astigmatism at 45 degrees & focus", "coma and x-tilt", "coma and y-tilt", "spherical & focus"};
```

```
Do[
 {tableData[i, 1] = wavefrontAberrationLabels[[i]], tableData[i, 2] = wavefrontAberrationList[[i]]}, {i, 1, 9}]
```

```
TableForm[Array[tableData, {9, 2}], TableHeadings → {{}, {"Aberration", "Zernike Term"}}]
```

Aberration	Zernike Term
piston	z_0
x-tilt	$\rho \, \text{Cos}[\theta] \, z_1$
y-tilt	$\rho \, \text{Sin}[\theta] \, z_2$
focus	$(-1 + 2 \rho^2) \, z_3$
astigmatism at 0 degrees & focus	$\rho^2 \, \text{Cos}[2 \theta] \, z_4$
astigmatism at 45 degrees & focus	$\rho^2 \, \text{Sin}[2 \theta] \, z_5$
coma and x-tilt	$\rho \, (-2 + 3 \rho^2) \, \text{Cos}[\theta] \, z_6$
coma and y-tilt	$\rho \, (-2 + 3 \rho^2) \, \text{Sin}[\theta] \, z_7$
spherical & focus	$(1 - 6 \rho^2 + 6 \rho^4) \, z_8$

The Zernike expansion above can be rewritten grouping like terms and equating them with the wavefront aberration coefficients.

```
wavefrontAberration = Collect[Sum[wavefrontAberrationList[[i]], {i, 9}], ρ]
```

$$z_0 - z_3 + \rho \, (\text{Cos}[\theta] \, z_1 + \text{Sin}[\theta] \, z_2 - 2 \, \text{Cos}[\theta] \, z_6 - 2 \, \text{Sin}[\theta] \, z_7) + \rho^3 \, (3 \, \text{Cos}[\theta] \, z_6 + 3 \, \text{Sin}[\theta] \, z_7) + \rho^2 \, (2 z_3 + \text{Cos}[2 \theta] \, z_4 + \text{Sin}[2 \theta] \, z_5 - 6 z_8) + z_8 + 6 \rho^4 \, z_8$$

```
piston = Select[wavefrontAberration, FreeQ[#, ρ] &];
```

```
tilt = Collect[Select[wavefrontAberration, MemberQ[#, ρ ] &], {ρ, Cos[θ], Sin[θ]}];
```

```
focusPlusAstigmatism = Select[wavefrontAberration, MemberQ[#, ρ^2 ] &];
```

```
coma = Select[wavefrontAberration, MemberQ[#, ρ^3 ] &];

spherical = Select[wavefrontAberration, MemberQ[#, ρ^4 ] &];
```

4.4 zernikeThirdOrderAberration Table

```
zernikeThirdOrderAberration = {{"piston", piston}, {"tilt", tilt},
   {"focus + astigmatism", focusPlusAstigmatism}, {"coma", coma}, {"spherical", spherical}};

TableForm[zernikeThirdOrderAberration]
```

piston	$z_0 - z_3 + z_8$
tilt	$\rho\,(\text{Cos}[\theta]\,(z_1 - 2\,z_6) + \text{Sin}[\theta]\,(z_2 - 2\,z_7))$
focus + astigmatism	$\rho^2\,(2\,z_3 + \text{Cos}[2\,\theta]\,z_4 + \text{Sin}[2\,\theta]\,z_5 - 6\,z_8)$
coma	$\rho^3\,(3\,\text{Cos}[\theta]\,z_6 + 3\,\text{Sin}[\theta]\,z_7)$
spherical	$6\,\rho^4\,z_8$

These tilt, coma, and focus plus astigmatism terms can be rearranged using the equation
$a\,\text{Cos}[\theta] + b\,\text{Sin}[\theta] = \sqrt{a^2 + b^2}\,\text{Cos}[\theta - \text{ArcTan}[a, b]]$.

4.4.1 Tilt

```
tilt = tilt /. a_ Cos[θ_] + b_ Sin[θ] → Sqrt[a^2 + b^2] Cos[θ - ArcTan[a, b]]
```

$$\rho\,\text{Cos}[\theta - \text{ArcTan}[z_1 - 2\,z_6,\ z_2 - 2\,z_7]]\,\sqrt{(z_1 - 2\,z_6)^2 + (z_2 - 2\,z_7)^2}$$

4.4.2 Coma

```
coma = Simplify[coma /. a_ Cos[θ_] + b_ Sin[θ] → Sqrt[a^2 + b^2] Cos[θ - ArcTan[a, b]]]
```

$$3\,\rho^3\,\text{Cos}[\theta - \text{ArcTan}[z_6, z_7]]\,\sqrt{z_6^2 + z_7^2}$$

4.4.3 Focus

This is a little harder because we must separate the focus and the astigmatism.

```
focusPlusAstigmatism
```

$$\rho^2\,(2\,z_3 + \text{Cos}[2\,\theta]\,z_4 + \text{Sin}[2\,\theta]\,z_5 - 6\,z_8)$$

```
focusPlusAstigmatism = focusPlusAstigmatism /. a_ Cos[θ_] + b_ Sin[θ_] → Sqrt[a^2 + b^2] Cos[θ - ArcTan[a, b]]
```

$$\rho^2\left(2\,z_3 + \text{Cos}[2\,\theta - \text{ArcTan}[z_4, z_5]]\,\sqrt{z_4^2 + z_5^2} - 6\,z_8\right)$$

But $\text{Cos}[2\,\phi] = 2\,\text{Cos}[\phi]^2 - 1$

```
focusPlusAstigmatism = focusPlusAstigmatism /. a_ Cos[2 θ_ - θ1_] → 2 a Cos[θ - θ1/2]^2 - a
```

$$\rho^2\left(2\,z_3 - \sqrt{z_4^2 + z_5^2} + 2\,\text{Cos}\left[\theta - \frac{1}{2}\,\text{ArcTan}[z_4, z_5]\right]^2\sqrt{z_4^2 + z_5^2} - 6\,z_8\right)$$

Let

$$\textbf{focusMinus} = \rho^2\left(2\,z_3 - \sqrt{z_4^2 + z_5^2} - 6\,z_8\right);$$

Sometimes $2\left(\sqrt{z_4{}^2 + z_5{}^2}\right)\rho^2$ is added to the focus term to make its absolute value smaller and then $2\left(\sqrt{z_4{}^2 + z_5{}^2}\right)\rho^2$ must be subtracted from the astigmatism term. This gives a focus term equal to

$$\textbf{focusPlus} = \rho^2\left(2\,z_3 + \sqrt{z_4^2 + z_5^2} - 6\,z_8\right);$$

For the focus we select the sign that will give the smallest magnitude.

```
focus = If[Abs[focusPlus / ρ^2] < Abs[focusMinus / ρ^2], focusPlus, focusMinus];
```

It should be noted that most commercial interferogram analysis programs do not try to minimize the absolute valus of the focus term so the focus is set equal to focusMinus.

4.4.4 Astigmatism

```
astigmatismMinus = focusPlusAstigmatism - focusMinus // Simplify
```

$$2\,\rho^2\,\text{Cos}\left[\theta - \frac{1}{2}\,\text{ArcTan}[z_4,\ z_5]\right]^2\,\sqrt{z_4^2 + z_5^2}$$

```
astigmatismPlus = focusPlusAstigmatism - focusPlus // Simplify
```

$$-2\,\rho^2\,\text{Sin}\left[\theta - \frac{1}{2}\,\text{ArcTan}[z_4,\ z_5]\right]^2\,\sqrt{z_4^2 + z_5^2}$$

Since $\text{Sin}[\theta - \frac{1}{2}\,\text{ArcTan}[z_4,\ z_5]]^2$ is equal to $\text{Cos}[\theta - (\frac{1}{2}\,\text{ArcTan}[z_4,\ z_5] + \frac{\pi}{2})]^2$, astigmatismPlus could be written as

$$\textbf{astigmatismPlus} = -2\,\rho^2\,\text{Cos}\left[\theta - \left(\frac{1}{2}\,\text{ArcTan}[z_4,\ z_5] + \frac{\pi}{2}\right)\right]^2\,\sqrt{z_4^2 + z_5^2};$$

Note that in going from astigmatismMinus to astigmatismPlus not only are we changing the sign of the astigmatism term, but we are also rotating it 90°.
We need to select the sign opposite that chosen in the focus term.

```
astigmatism = If[Abs[focusPlus / ρ^2] < Abs[focusMinus / ρ^2], astigmatismPlus, astigmatismMinus];
```

Again it should be noted that most commercial interferogram analysis programs do not try to minimize the absolute valus of the focus term and the astigmatism is given by astigmatismMinus.

4.4.5 Spherical

$$\textbf{spherical} = 6\,z_8\,\rho^4$$

4.5 seidelAberrationList Table

We can summarize the results as follows.

```
seidelAberrationList := {{"piston", piston}, {"tilt", tilt},
   {"focus", focus}, {"astigmatism", astigmatism}, {"coma", coma}, {"spherical", spherical}};

seidelAberrationList // TableForm
```

piston	$z_0 - z_3 + z_8$
tilt	$\rho\,\mathrm{Cos}[\theta - \mathrm{ArcTan}[z_1 - 2 z_6, z_2 - 2 z_7]]\sqrt{(z_1 - 2 z_6)^2 + (z_2 - 2 z_7)^2}$
focus	$\mathrm{If}\left[\mathrm{Abs}\left[2 z_3 + \sqrt{z_4^2 + z_5^2} - 6 z_8\right] < \mathrm{Abs}\left[2 z_3 - \sqrt{z_4^2 + z_5^2} - 6 z_8\right], \mathrm{focusPlus}, \mathrm{focusMinus}\right]$
astigmatism	$\mathrm{If}\left[\mathrm{Abs}\left[2 z_3 + \sqrt{z_4^2 + z_5^2} - 6 z_8\right] < \mathrm{Abs}\left[2 z_3 - \sqrt{z_4^2 + z_5^2} - 6 z_8\right], \mathrm{astigmatismPlus}, \mathrm{astigmatismMinus}\right]$
coma	$3\rho^3\,\mathrm{Cos}[\theta - \mathrm{ArcTan}[z_6, z_7]]\sqrt{z_6^2 + z_7^2}$
spherical	$6\rho^4 z_8$

4.5.1 Typical Results

```
seidelAberrationList //. {z0 → 0, z1 → 1, z2 → 1, z3 → 1, z4 → 3, z5 → 1, z6 → 1, z7 → 1, z8 → 2} // TableForm
```

piston	1
tilt	$\sqrt{2}\,\rho\,\mathrm{Cos}\left[\frac{3\pi}{4} + \theta\right]$
focus	$\left(-10 + \sqrt{10}\right)\rho^2$
astigmatism	$-2\sqrt{10}\,\rho^2\,\mathrm{Sin}\left[\theta - \frac{1}{2}\mathrm{ArcTan}\left[\frac{1}{3}\right]\right]^2$
coma	$3\sqrt{2}\,\rho^3\,\mathrm{Cos}\left[\frac{\pi}{4} - \theta\right]$
spherical	$12\rho^4$

```
seidelAberration = Apply[Plus, seidelAberrationList][[2]];

seidelAberration //. {z0 → 0, z1 → 1, z2 → 1, z3 → 1, z4 → 3, z5 → 1, z6 → 1, z7 → 1, z8 → 2}
```

$$1 + \left(-10 + \sqrt{10}\right)\rho^2 + 12\rho^4 + 3\sqrt{2}\,\rho^3\,\mathrm{Cos}\left[\frac{\pi}{4} - \theta\right] + \sqrt{2}\,\rho\,\mathrm{Cos}\left[\frac{3\pi}{4} + \theta\right] - 2\sqrt{10}\,\rho^2\,\mathrm{Sin}\left[\theta - \frac{1}{2}\mathrm{ArcTan}\left[\frac{1}{3}\right]\right]^2$$

5 RMS Wavefront Aberration

If the wavefront aberration can be described in terms of third-order aberrations, it is convenient to specify the wavefront aberration by stating the number of waves of each of the third-order aberrations present. This method for specifying a wavefront is of particular convenience if only a single third-order aberration is present. For more complicated wavefront aberrations it is convenient to state the peak-to-valley (P-V) sometimes called peak-to-peak (P-P) wavefront aberration. This is simply the maximum departure of the actual wavefront from the desired wavefront in both positive and negative directions. For example, if the maximum departure in the positive direction is +0.2 waves and the maximum departure in the negative direction is -0.1 waves, then the P-V wavefront error is 0.3 waves.

While using P-V to specify wavefront error is convenient and simple, it can be misleading. Stating P-V is simply stating the maximum wavefront error, and it is telling nothing about the area over which this error is occurring. An optical system having a large P-V error may actually perform better than a system having a small P-V error. It is generally more meaningful to specify wavefront quality using the rms wavefront error.

The next equation defines the rms wavefront error σ for a circular pupil, as well as the variance σ^2. $\Delta w(\rho, \theta)$ is measured relative to the best fit spherical wave, and it generally has the units of waves. $\overline{\Delta w}$ is the mean wavefront OPD.

$$\text{average}[\Delta w\_] := \frac{1}{\pi} \int_0^{2\pi} \int_0^1 \Delta w \, \rho \, d\rho \, d\theta;$$

$$\text{standardDeviation}[\Delta w\_] := \sqrt{\frac{1}{\pi} \int_0^{2\pi} \int_0^1 (\Delta w - \text{average}[\Delta w])^2 \rho \, d\rho \, d\theta}$$

As an example we will calculate the relationship between σ and the mean wavefront aberrations for the third-order aberrations of a circular pupil.

```
meanRmsList = {{"Defocus", "w20 ρ^2", w20 average[ρ^2], w20 N[standardDeviation[ρ^2], 3]},
   {"Spherical", "w40 ρ^4", w40 average[ρ^4], w40 N[standardDeviation[ρ^4], 3]},
   {"Spherical & Defocus", "w40 (ρ^4-ρ^2)", w40 average[(ρ^4 - ρ^2)], w20 N[standardDeviation[(ρ^4 - ρ^2)], 3]},
   {"Astigmatism", "w22 ρ^2 Cos[θ]^2", w22 average[ρ^2 Cos[θ]^2], w22 N[standardDeviation[ρ^2 Cos[θ]^2], 3]},
   {"Astigmatism & Defocus", "w22 ρ^2 (Cos[θ]^2-1/2)",
    w22 average[ρ^2 (Cos[θ]^2 - 1/2)], w22 N[standardDeviation[ρ^2 (Cos[θ]^2 - 1/2)], 3]},
   {"Coma", "w31 ρ^3 Cos[θ]", w31 average[ρ^3 Cos[θ]], w31 N[standardDeviation[ρ^3 Cos[θ]], 3]}, {"Coma & Tilt",
    "w31 (ρ^3-2/3 ρ) Cos[θ]", w31 average[(ρ^3 - 2/3 ρ) Cos[θ]], w31 N[standardDeviation[(ρ^3 - 2/3 ρ) Cos[θ]], 3]}};
```

```
TableForm[meanRmsList, TableHeadings -> {{}, {"Aberration", "Δw", "Δw̄", "RMS"}}]
```

Aberration	Δw	$\overline{\Delta w}$	RMS
Defocus	$w_{20}\,\rho^2$	$\frac{w_{20}}{2}$	$0.288675\,w_{20}$
Spherical	$w_{40}\,\rho^4$	$\frac{w_{40}}{3}$	$0.298142\,w_{40}$
Spherical & Defocus	$w_{40}\,(\rho^4-\rho^2)$	$-\frac{w_{40}}{6}$	$0.0745356\,w_{20}$
Astigmatism	$w_{22}\,\rho^2\,\mathrm{Cos}[\theta]^2$	$\frac{w_{22}}{4}$	$0.25\,w_{22}$
Astigmatism & Defocus	$w_{22}\,\rho^2\,(\mathrm{Cos}[\theta]^2-\frac{1}{2})$	0	$0.204124\,w_{22}$
Coma	$w_{31}\,\rho^3\,\mathrm{Cos}[\theta]$	0	$0.353553\,w_{31}$
Coma & Tilt	$w_{31}\,(\rho^3-\frac{2}{3}\rho)\,\mathrm{Cos}[\theta]$	0	$0.117851\,w_{31}$

If the wavefront aberration can be expressed in terms of Zernike polynomials, the wavefront variance can be calculated in a simple form by using the orthogonality relations of the Zernike polynomials. The final result for the entire unit circle is

$$\sigma = \sqrt{\sum_{n=1}^{\mathrm{nmax}}\left(\frac{a[n]^2}{2\,n+1}+\frac{1}{2}\sum_{m=1}^{n}\frac{b[n,\,m]^2+c[n,\,m]^2}{2\,n+1-m}\right)};$$

The following table gives the relationship between σ and the Zernike polynomials if the Zernike coefficients are unity.

```
zernikeRms = Table[If[zernikePolarList[[i, 3]] == 0, 1/Sqrt[2 zernikePolarList[[i, 2]] + 1],
    1/Sqrt[2 (2 zernikePolarList[[i, 2]] + 1 - zernikePolarList[[i, 3]])]], {i, Length[zernikePolarList]}];

zernikePolarRmsList = Transpose[Insert[Transpose[zernikePolarList], zernikeRms, 4]];

TableForm[zernikePolarRmsList, TableHeadings -> {{}, {"#", "n", "m", "RMS", "Polynomial"}}]
```

#	n	m	RMS	Polynomial
0	0	0	1	1
1	1	1	$\frac{1}{2}$	$\rho\,\mathrm{Cos}[\theta]$
2	1	1	$\frac{1}{2}$	$\rho\,\mathrm{Sin}[\theta]$
3	1	0	$\frac{1}{\sqrt{3}}$	$-1+2\,\rho^2$

4	2	2	$\frac{1}{\sqrt{6}}$	ρ^2 Cos[2θ]
5	2	2	$\frac{1}{\sqrt{6}}$	ρ^2 Sin[2θ]
6	2	1	$\frac{1}{2\sqrt{2}}$	$\rho\,(-2+3\rho^2)$ Cos[θ]
7	2	1	$\frac{1}{2\sqrt{2}}$	$\rho\,(-2+3\rho^2)$ Sin[θ]
8	2	0	$\frac{1}{\sqrt{5}}$	$1-6\rho^2+6\rho^4$
9	3	3	$\frac{1}{2\sqrt{2}}$	ρ^3 Cos[3θ]
10	3	3	$\frac{1}{2\sqrt{2}}$	ρ^3 Sin[3θ]
11	3	2	$\frac{1}{\sqrt{10}}$	$\rho^2\,(-3+4\rho^2)$ Cos[2θ]
12	3	2	$\frac{1}{\sqrt{10}}$	$\rho^2\,(-3+4\rho^2)$ Sin[2θ]
13	3	1	$\frac{1}{2\sqrt{3}}$	$\rho\,(3-12\rho^2+10\rho^4)$ Cos[θ]
14	3	1	$\frac{1}{2\sqrt{3}}$	$\rho\,(3-12\rho^2+10\rho^4)$ Sin[θ]
15	3	0	$\frac{1}{\sqrt{7}}$	$-1+12\rho^2-30\rho^4+20\rho^6$
16	4	4	$\frac{1}{\sqrt{10}}$	ρ^4 Cos[4θ]
17	4	4	$\frac{1}{\sqrt{10}}$	ρ^4 Sin[4θ]
18	4	3	$\frac{1}{2\sqrt{3}}$	$\rho^3\,(-4+5\rho^2)$ Cos[3θ]
19	4	3	$\frac{1}{2\sqrt{3}}$	$\rho^3\,(-4+5\rho^2)$ Sin[3θ]
20	4	2	$\frac{1}{\sqrt{14}}$	$\rho^2\,(6-20\rho^2+15\rho^4)$ Cos[2θ]
21	4	2	$\frac{1}{\sqrt{14}}$	$\rho^2\,(6-20\rho^2+15\rho^4)$ Sin[2θ]
22	4	1	$\frac{1}{4}$	$\rho\,(-4+30\rho^2-60\rho^4+35\rho^6)$ Cos[θ]
23	4	1	$\frac{1}{4}$	$\rho\,(-4+30\rho^2-60\rho^4+35\rho^6)$ Sin[θ]
24	4	0	$\frac{1}{3}$	$1-20\rho^2+90\rho^4-140\rho^6+70\rho^8$
25	5	5	$\frac{1}{2\sqrt{3}}$	ρ^5 Cos[5θ]
26	5	5	$\frac{1}{2\sqrt{3}}$	ρ^5 Sin[5θ]
27	5	4	$\frac{1}{\sqrt{14}}$	$\rho^4\,(-5+6\rho^2)$ Cos[4θ]
28	5	4	$\frac{1}{\sqrt{14}}$	$\rho^4\,(-5+6\rho^2)$ Sin[4θ]
29	5	3	$\frac{1}{4}$	$\rho^3\,(10-30\rho^2+21\rho^4)$ Cos[3θ]
30	5	3	$\frac{1}{4}$	$\rho^3\,(10-30\rho^2+21\rho^4)$ Sin[3θ]

31	5	2	$\frac{1}{3\sqrt{2}}$	$\rho^2(-10+60\rho^2-105\rho^4+56\rho^6)$ Cos[2θ]
32	5	2	$\frac{1}{3\sqrt{2}}$	$\rho^2(-10+60\rho^2-105\rho^4+56\rho^6)$ Sin[2θ]
33	5	1	$\frac{1}{2\sqrt{5}}$	$\rho(5-60\rho^2+210\rho^4-280\rho^6+126\rho^8)$ Cos[θ]
34	5	1	$\frac{1}{2\sqrt{5}}$	$\rho(5-60\rho^2+210\rho^4-280\rho^6+126\rho^8)$ Sin[θ]
35	5	0	$\frac{1}{\sqrt{11}}$	$-1+30\rho^2-210\rho^4+560\rho^6-630\rho^8+252\rho^{10}$
36	6	6	$\frac{1}{\sqrt{14}}$	ρ^6 Cos[6θ]
37	6	6	$\frac{1}{\sqrt{14}}$	ρ^6 Sin[6θ]
38	6	5	$\frac{1}{4}$	$\rho^5(-6+7\rho^2)$ Cos[5θ]
39	6	5	$\frac{1}{4}$	$\rho^5(-6+7\rho^2)$ Sin[5θ]
40	6	4	$\frac{1}{3\sqrt{2}}$	$\rho^4(15-42\rho^2+28\rho^4)$ Cos[4θ]
41	6	4	$\frac{1}{3\sqrt{2}}$	$\rho^4(15-42\rho^2+28\rho^4)$ Sin[4θ]
42	6	3	$\frac{1}{2\sqrt{5}}$	$\rho^3(-20+105\rho^2-168\rho^4+84\rho^6)$ Cos[3θ]
43	6	3	$\frac{1}{2\sqrt{5}}$	$\rho^3(-20+105\rho^2-168\rho^4+84\rho^6)$ Sin[3θ]
44	6	2	$\frac{1}{\sqrt{22}}$	$\rho^2(15-140\rho^2+420\rho^4-504\rho^6+210\rho^8)$ Cos[2θ]
45	6	2	$\frac{1}{\sqrt{22}}$	$\rho^2(15-140\rho^2+420\rho^4-504\rho^6+210\rho^8)$ Sin[2θ]
46	6	1	$\frac{1}{2\sqrt{6}}$	$\rho(-6+105\rho^2-560\rho^4+1260\rho^6-1260\rho^8+462\rho^{10})$ Cos[θ]
47	6	1	$\frac{1}{2\sqrt{6}}$	$\rho(-6+105\rho^2-560\rho^4+1260\rho^6-1260\rho^8+462\rho^{10})$ Sin[θ]
48	6	0	$\frac{1}{\sqrt{13}}$	$1-42\rho^2+420\rho^4-1680\rho^6+3150\rho^8-2772\rho^{10}+924\rho^{12}$

6 Strehl Ratio

While in the absence of aberrations, the intensity is a maximum at the Gaussian image point, if aberrations are present this will in general no longer be the case. The point of maximum intensity is called diffraction focus, and for small aberrations is obtained by finding the appropriate amount of tilt and defocus to be added to the wavefront so that the wavefront variance is a minimum.

The ratio of the intensity at the Gaussian image point (the origin of the reference sphere is the point of maximum intensity in the observation plane) in the presence of aberration, divided by the intensity that would be obtained if no aberration were present, is called the Strehl ratio, the Strehl definition, or the Strehl intensity. The Strehl ratio is given by

$$\text{strehlRatio} := \frac{1}{\pi^2}\,\text{Abs}\left[\int_0^{2\pi}\int_0^1 e^{i\,2\pi\,\Delta w[\rho,\theta]}\,\rho\,d\rho\,d\theta\right]^2$$

where $\Delta w[\rho, \theta]$ in units of waves. As an example

```
strehlRatio /. Δw[ρ, θ] → ρ^3 Cos[θ] // N
```

```
0.0790649
```

where $\Delta w[\rho, \theta]$ is in units of waves. The above equation may be expressed in the form

$$\text{strehlRatio} = \frac{1}{\pi^2}\,\text{Abs}\left[\int_0^{2\pi}\int_0^1 \left(1 + 2\,i\,\pi\,\Delta w[\rho, \theta] - 2\,\pi^2\,\Delta w[\rho, \theta]^2 + \cdots\right)\rho\,d\rho\,d\theta\right]^2$$

If the aberrations are so small that the third-order and higher-order powers of $2\pi\Delta w$ can be neglected, the above equation may be written as

$$\begin{aligned}\text{strehlRatio} &\approx \text{Abs}\left[1 + i2\pi\,\overline{\Delta w} - \frac{1}{2}\,(2\,\pi)^2\,\overline{\Delta w^2}\right]^2 \\ &\approx 1 - (2\,\pi)^2\left(\overline{\Delta w^2} - \overline{(\Delta w)}^2\right) \\ &\approx 1 - (2\,\pi\,\sigma)^2\end{aligned}$$

where σ is in units of waves.

Thus, when the aberrations are small, the Strehl ratio is independent of the nature of the aberration and is smaller than the ideal value of unity by an amount proportional to the variance of the wavefront deformation.

The above equation is valid for Strehl ratios as low as about 0.5. The Strehl ratio is always somewhat larger than would be predicted by the above approximation. A better approximation for most types of aberration is given by

$$\text{strehlRatioApproximation} := e^{-(2\pi\sigma)^2}$$

$$\text{strehlRatioApproximation} \approx 1 - (2\,\pi\sigma)^2 + \frac{(2\,\pi\sigma)^4}{2} + \cdots$$

which is good for Strehl ratios as small as 0.1.

Once the normalized intensity at diffraction focus has been determined, the quality of the optical system may be ascertained using the Marechal criterion. The Marecha1 criterion states that a system is regarded as well corrected if the normalized intensity at diffraction focus is greater than or equal to 0.8, which corresponds to an rms wavefront error $<\lambda/14$.

As mentioned above, a useful feature of Zernike polynomials is that each term of the Zernikes minimizes the rms wavefront error to the order of that term. That is, each term is structured such that adding other aberrations of lower orders can only increase the rms error. Removing the first-order Zernike terms of tilt and defocus represents a shift in the focal point that maximizes the intensity at that point. Likewise, higher order terms have built into them the appropriate amount of tilt and defocus to minimize the rms wavefront error to that order. For example, looking at Zernike term #9 shows that for each wave of third-order spherical aberration present, one wave of defocus should be subtracted to minimize the rms wavefront error and find diffraction focus.

7 References

Born, M. and Wolf, E., (1959). Principles of Optics, pp. 464-466, 767-772. Pergamon press, New York.
Kim, C.-J. and Shannon, R.R. (1987). In "Applied Optics and Optical Engineering," Vol. X (R. Shannon and J. Wyant, eds.), pp. 193-221. Academic Press, New York.
Wyant, J. C. and Creath, K. (1992). In "Applied Optics and Optical Engineering," Vol. XI (R. Shannon and J. Wyant, eds.), pp. 28-39. Academic Press, New York.
Zernike, F. (1934), *Physica* **1**, 689.

8 Index